



● 产品名称

MK7XXXX系列芯片

● 标题

关于MK7XXXX系列芯片的RC测温

● 简介

目前市场上许多产品需要进行温度控制。出于价格的考虑，用户大都采用电容冲放电的办法来实现测温的目的。温度传感器一般采用热敏电阻。这种传感器的特性是温度越高，阻值越低，温度越低，阻值越高。

图 1 是一个 MK7A10P RC 测温试验电路图。基本功能要求是测温，并将测得的温度通过 PortB 端口显示出来。




图 1 MK7A10P RC 测温试验

● DEMO 程序

➤ 汇编程序文档

```
-----  
#include "mk7a10p_hw.inc" ;编译该文档需包含"mk7a10p_hw.inc"文件  
-----  
;芯片型号 (mk7a10p)
```



```
;-----
;配置寄存器设置说明 (CONFIG)
;1---FOSC=NS                      ;LS,NS,HS,RC
;2---WDTE=Enable                  ;Enable,Disable
;3---CPT=OFF                      ;ON,OFF
;4---LV=2v                        ;4V,2V,Don't use
;5---RTCEN=PA4 input              ;Timer input only,PA4 input
;6---TYPE=... has no ... wake up ... ;... has no ... wake up ...
;... has ... wake up ...

;-----
rt_r0      equ      0x07          ;rt 充电计数值低位
rt_r1      equ      0x08          ;rt 充电计数值高位
ct_r0      equ      0x09          ;ct 充电计数值低位
ct_r1      equ      0x0a          ;ct 充电计数值高位
buffer0     equ      0x0b          ;缓存寄存器 0
buffer1     equ      0x0c          ;缓存寄存器 1
flag        equ      0x0d          ;标志寄存器
count       equ      0x0e
tm_r        equ      0x0f
div0        equ      0x10
div1        equ      0x11
val_r       equ      0x12

;-----
#define      flag_err  flag,0      ;错误标志
#define      flag_sel  flag,1      ;检测对象选择标志
;-----
;=====
org          0x3ff                ;mk7a10p 的复位向量地址定义
lgoto       main                  ;跳转到主程序入口
org          0x000

;=====
;rt/tc 的高位表, 注释中 R 为传感器的阻值,这里使用: AT-103 型传感器
;=====
temp_h
add          pc,m
retla       .56                   ;"0'c"-----R:27.28K
retla       .53                   ;"1'c"-----R:26.13K
retla       .51                   ;"2'c"-----R:25.03K
retla       .49                   ;"3'c"-----R:24.00K
retla       .47                   ;"4'c"-----R:23.00K
```



retla	.45	;"5'c"-----R:22.05K
retla	.43	;"6'c"-----R:21.15K
retla	.41	;"7'c"-----R:20.30K
retla	.40	;"8'c"-----R:19.48K
retla	.38	;"9'c"-----R:18.70K
retla	.37	;"10'c"-----R:17.96K
retla	.35	;"11'c"-----R:17.24K
retla	.34	;"12'c"-----R:16.65K
retla	.32	;"13'c"-----R:15.90K
retla	.31	;"14'c"-----R:15.28K
retla	.30	;"15'c"-----R:14.69K
retla	.29	;"16'c"-----R:14.12K
retla	.28	;"17'c"-----R:13.58K
retla	.27	;"18'c"-----R:13.06K
retla	.26	;"19'c"-----R:12.56K
retla	.25	;"20'c"-----R:12.09K
retla	.24	;"21'c"-----R:11.63K
retla	.23	;"22'c"-----R:11.20K
retla	.22	;"23'c"-----R:10.78K
retla	.21	;"24'c"-----R:10.38K
retla	.20	;"25'c"-----R:10.00K
retla	.19	;"26'c"-----R:9.632K
retla	.19	;"27'c"-----R:9.281K
retla	.18	;"28'c"-----R:8.944K
retla	.17	;"29'c"-----R:8.622K
retla	.17	;"30'c"-----R:8.313K
retla	.16	;"31'c"-----R:8.014K
retla	.16	;"32'c"-----R:7.728K
retla	.15	;"33'c"-----R:7.454K
retla	.14	;"34'c"-----R:7.192K
retla	.14	;"35'c"-----R:6.940K
retla	.13	;"36'c"-----R:6.699K
retla	.13	;"37'c"-----R:6.467K
retla	.12	;"38'c"-----R:6.245K
retla	.12	;"39'c"-----R:6.032K
retla	.12	;"40'c"-----R:5.827K

=====

;rt/tc 的低位表, 注释中 R 为传感器的阻值,这里使用: AT-103 型传感器

=====

temp_l



add	pc,m	
retla	.15	;"0'c"-----R:27.28K
retla	.74	;"1'c"-----R:26.13K
retla	.61	;"2'c"-----R:25.03K
retla	.48	;"3'c"-----R:24.00K
retla	.15	;"4'c"-----R:23.00K
retla	.32	;"5'c"-----R:22.05K
retla	.68	;"6'c"-----R:21.15K
retla	.79	;"7'c"-----R:20.30K
retla	.02	;"8'c"-----R:19.48K
retla	.50	;"9'c"-----R:18.70K
retla	.03	;"10'c"-----R:17.96K
retla	.68	;"11'c"-----R:17.24K
retla	.27	;"12'c"-----R:16.65K
retla	.84	;"13'c"-----R:15.90K
retla	.61	;"14'c"-----R:15.28K
retla	.32	;"15'c"-----R:14.69K
retla	.19	;"16'c"-----R:14.12K
retla	.17	;"17'c"-----R:13.58K
retla	.12	;"18'c"-----R:13.06K
retla	.10	;"19'c"-----R:12.56K
retla	.02	;"20'c"-----R:12.09K
retla	.04	;"21'c"-----R:11.63K
retla	.20	;"22'c"-----R:11.20K
retla	.30	;"23'c"-----R:10.78K
retla	.40	;"24'c"-----R:10.38K
retla	.68	;"25'c"-----R:10.00K
retla	.93	;"26'c"-----R:9.632K
retla	.34	;"27'c"-----R:9.281K
retla	.68	;"28'c"-----R:8.944K
retla	.78	;"29'c"-----R:8.622K
retla	.42	;"30'c"-----R:8.313K
retla	.85	;"31'c"-----R:8.014K
retla	.01	;"32'c"-----R:7.728K
retla	.44	;"33'c"-----R:7.454K
retla	.93	;"34'c"-----R:7.192K
retla	.40	;"35'c"-----R:6.940K
retla	.95	;"36'c"-----R:6.699K
retla	.50	;"37'c"-----R:6.467K
retla	.96	;"38'c"-----R:6.245K



```
retla      .45      ;"39'c"-----R:6.032K
retla      .05      ;"40'c"-----R:5.827K
;=====
main
    clr     0x07
    clr     0x08
    clr     0x09
    clr     0x0a
    clr     0x0b
    clr     0x0c
    clr     0x0d
    clr     0x0e
    clr     0x0f
    clr     0x10
    clr     0x11
    clr     0x12
    clr     0x13
    ;-----
    movla   b'11110000'
    iodir    porta      ;PortA0-PortA3 为 output
    movla   b'00000000'
    iodir    portb      ;PortB0-PortB7 为 output
    clr     porta
    clr     portb
    ;-----
    movla   b'00000010'      ;配置 TMR0,预分
    select
    clr     tmr0
;=====
read_tm
    lrwdt
    movla   .250
    xor     tmr0,a
    btss    status,z
    lgoto   $-4
    clr     tmr0
    ;-----
    inc     tm_r,m
    movla   .250
    xor     tm_r,a
```



```
btss      status,z
lgoto     tm_r
;=====
test_temp
lcall     test_oc      ;调用测量充电时间子程序
;-----
btss      flag_err
lgoto     $+9
;-----
;错误出现所有 LED 闪烁
bc        flag_err
movla     b'10000000'
xor       flag,m
movla     0x00
btsc      flag,7
movla     0xff
movam     portb
lgoto     read_tm
;-----
btsc      flag_sel
lgoto     read_tm
;-----
;测温数据处理: (rt_r0,rt_r1) / (ct_r0,ct_r1)
;相除后结果放到 div1 和 div0 中
;其中 div1 的高 4bit 放整数部分, 低 4bit 放小数部分的十分
;div0 的高 4bit 放小数部分的百分位, 低 4bit 放小数部分的
;-----
clr       div0
clr       div1
clr       count
;-----
div_0
;rt 值/ct 值,结果存入 div1 高半字节
lcall     sub16bit
btss      status,c
lgoto     div_1      ;不够减
movla     0x10      ;div1 高半字节加 1
add       div1,m
lgoto     div_0
;-----
```



div_1

lcall multi_10 ;(rt_r1,rt_r0)x10

div_2

;rt 值/ct 值,结果存入 div1 低半字节

lcall sub16bit

btss status,c

lgoto div_3 ;不够减

movla 0x01 ;div1 低半字节加 1

add div1,m

lgoto div_2

div_3

lcall multi_10 ;(rt_r1,rt_r0)x10

div_4

;rt 值/ct 值,结果存入 div1 低半字节

lcall sub16bit

btss status,c

lgoto div_5 ;不够减

movla 0x10 ;div0 高半字节加 1

add div0,m

lgoto div_4

div_5

lcall multi_10 ;(rt_r1,rt_r0)x10

div_6

;rt 值/ct 值,结果存入 div1 低半字节

lcall sub16bit

btss status,c

lgoto look_temp ;不够减

movla 0x01 ;div0 低半字节加 1

add div0,m

lgoto div_6

look_temp

clr val_r

look_lp



```
mov    val_r,a
lcall  temp_h
movam  ct_r1
mov    val_r,a
lcall  temp_l
movam  ct_r0
;-----
mov    div1,a
movam  rt_r1
mov    div0,a
movam  rt_r0
;-----
lcall  sub16bit          ;仅仅比较大小
;-----
btsc   status,c
lgoto  get_val
;-----
inc    val_r,m
movla  .40
xor    val_r,a
btss   status,z
lgoto  look_lp
;-----
get_val
mov    val_r,a
movam  portb            ;将所得的值送 PortB 显示
lgoto  read_tm
;=====
;充电时间测量子程序
;=====
test_oc
btsc   flag_sel
lgoto  test_rt
;-----
test_ct
clr    ct_r0
clr    ct_r1
movla  b'11110100'
iodir  porta
bs     porta,3
```




```
;-----  
ct_loop  
    incsz    ct_r0,m  
    lgoto    $+4  
    incsz    ct_r1,m  
    lgoto    $+2  
    lgoto    test_err  
;-----  
    btss     porta,2  
    lgoto    ct_loop  
;-----  
    bs       flag_sel  
    lgoto    test_end  
;-----  
test_rt  
    clr      rt_r0  
    clr      rt_r1  
    movla    b'11111000'  
    iodir    porta  
    bs       porta,2  
;-----  
rt_loop  
    incsz    rt_r0,m  
    lgoto    $+4  
    incsz    rt_r1,m  
    lgoto    $+2  
    lgoto    test_err  
;-----  
    btss     porta,3  
    lgoto    rt_loop  
;-----  
    bc       flag_sel  
    lgoto    test_end  
;-----  
test_end  
    movla    b'11110000'  
    iodir    porta  
    clr      porta  
    ret  
;-----
```



test_err

bs flag_err ;置传感器出错标志
bc flag_sel
ret

=====

;16bit 无符号二进制减法子程序,不够减就退出

=====

sub16bit

mov ct_r1,a
sub rt_r1,a
btss status,c
ret ;不够减,原寄存器的值不变
btsc status,z
lgoto \$+9

movam rt_r1
mov ct_r0,a
sub rt_r0,a
movam rt_r0
btss status,c
dec rt_r1,m
bs status,c
ret ;够减,结果存入 rt_r1 和 rt_r0

mov ct_r0,a
sub rt_r0,a
btss status,c
ret ;不够减,原寄存器的值不变
movam rt_r0
clr rt_r1
bs status,c
ret ;够减,结果存入 rt_r1 和 rt_r0

=====

;16bit 二进制乘 10 子程序

=====

multi_10

movla .9
movam count
mov rt_r0,a
movam buffer0



```
mov      rt_r1,a
movam    buffer1
;-----
add_loop
mov      buffer0,a
add      rt_r0,m
btsc     status,c
inc      rt_r1,m
mov      buffer1,a
add      rt_r1,m
decsz    count,m
lgoto    add_loop
ret
;=====
end
```

➤ mk7a10p_hw.inc 文档

```
;-----Define special register(Define SFR)-----
indf      equ      0x00
tmr0      equ      0x01
pc         equ      0x02
status     equ      0x03
fsr        equ      0x04
porta     equ      0x05      ;porta(0-3)
portb     equ      0x06      ;portb(0-7)
;-----Define [status Register] special bit-----
c          equ      0
dc         equ      1
z          equ      2
pd         equ      3
to         equ      4
sa0        equ      5
;-----
```